

wxMaxima pour l'Analyse Volume II

Auteur : Zachary Hannan, Solano Community College
Les corrections sont réalisées par Michel Gosse

Exercices du Chapitre 3 et leur corrigé

Présentation

Le manuel wxMaxima for Calculus Volume II, dont l'auteur est Zachary Hannan, propose l'étude de nombreux domaines de l'analyse à l'aide du logiciel Maxima et de son interface graphique wxMaxima. A l'occasion de la traduction française de ce manuel, la correction des exercices de chaque chapitre est proposée. Ce document reprend les exercices du manuel et leur correction à l'aide de wxMaxima.

La dernière version de ce document, ainsi que les fichiers source pour wxMaxima, sont téléchargeables sur le site <https://maxima-french-doc.fr/>, à la rubrique Documentation.

Enoncé des exercices

1. Calculer une approximation décimale de l'aire algébrique délimitée par $f(x) = -x^2 + 2x$ sur $[0, 4]$. Réaliser un graphique en colorant les aires correspondantes, en mettant en rouge la contribution associée à l'aire algébrique positive et en bleu celle de l'aire algébrique négative.
2. Calculer une approximation décimale de l'aire délimitée entre $f(x) = \sqrt{1-x^2}$ et $g(x) = \cos(\frac{\pi}{2}x)$. Vous pouvez utiliser la symétrie pour simplifier votre calcul, mais vous devez inclure l'argument algébrique approprié pour justifier l'utilisation de la symétrie. Produire un graphique coloré illustrant l'aire calculée.
3. Utiliser un graphique pour étudier les courbes de $f(x) = (x-3)(x-2)(x-1)x(x+1)(x+2)(x+3)$ et $g(x) = e^x$. Calculer avec wxMaxima toutes les intersections de ces deux fonctions, puis trouver l'aire délimitée entre elles. Comme dans l'Exemple 3.1.3, il faut veiller à rendre chaque contribution d'aire positive en choisissant la bonne différence entre les fonctions dans chaque intégrale d'aire.
4. Calculez le volume du solide obtenu par révolution de $f(x) = \cos x$ sur $[0, \pi]$ autour de l'axe des y .

Remarque : puisque nous travaillons sur le domaine de restriction « classique » de la fonction cosinus, on peut ignorer l'avertissement « certaines solutions peuvent être perdues ».

- (a) Utilisez la méthode des disques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - (b) Utilisez la méthode des coques cylindriques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - (c) * Utilisez `wxdraw3d` pour tracer le solide ainsi que chaque type d'élément de volume.
5. Reprenez le dernier exemple en utilisant cette fois une révolution autour de l'axe des x au lieu de l'axe des y .
 6. Tracez le graphique ombré de la région délimitée par $f(x) = 1 + 0,5x$ et $g(x) = x$ sur $[0, 2]$.
 - (a) Utilisez la méthode des rondelles pour calculer le volume du solide obtenu par révolution de cette région autour de l'axe des y .
 - (b) Utilisez la méthode des rondelles pour calculer le volume du solide obtenu par révolution de cette région autour de l'axe des x .
 - (c) * Utilisez `wxdraw3d` pour tracer chaque solide, en incluant un élément de volume.

7. Tracez le graphique ombré de la région délimitée par $f(x) = \sin x$ et $g(x) = \cos x$ entre leurs deux premières intersections sur $[0, \frac{3\pi}{2}]$. Utilisez la méthode des rondelles pour calculer le volume du solide obtenu par révolution de cette région autour de l'axe des y .
- Remarque* : bien que le cosinus soit restreint au domaine classique $[0, \pi]$, la fonction sinus ne l'est *pas*. Vous devrez donc réfléchir attentivement à la façon d'ajuster le rayon extérieur sur $[\frac{\pi}{2}, \frac{3\pi}{2}]$.
- Vérifiez que la même réponse est obtenue par la méthode des coques cylindriques.
8. Calculer la longueur d'arc de la fonction $f(x) = \sin x$ sur $[0, 2\pi]$ en utilisant deux méthodes différentes :
- Utiliser un processus limite avec $n = 4, 8, 16, 32, 64$ sous-intervalles, en incluant les graphiques de chaque approximation.
 - Utiliser la formule intégrale de la longueur d'arc pour calculer une approximation décimale.
 - Calculer l'erreur relative (en pourcentage) en comparant l'approximation pour $n = 64$ et la méthode intégrale.
9. Calculer l'aire de la surface obtenue en faisant tourner la courbe représentative de $f(x) = e^{-0.2x} \cos 3x$ sur $[0, 10]$ autour de l'axe des x . Tracer la surface avec `wxdraw3d`.
10. Calculer l'aire de la surface obtenue en faisant tourner la courbe représentative de $f(x) = e^{-0.2x} \cos 3x$ sur $[0, 10]$ autour de l'axe des y . Tracer la surface avec `wxdraw3d`.
11. Calculer l'aire d'un disque de rayon R de deux façons différentes :
- Utiliser l'équation du demi-cercle supérieur et calculer directement l'aire en intégrant par rapport à x .
 - Utiliser des anneaux concentriques comme éléments d'aire. Chaque anneau doit avoir un rayon r et une épaisseur dr . « Dérouler » un anneau et calculer l'élément d'aire dA , puis intégrer pour sommer les éléments d'aire.
12. Calculez le volume d'un cylindre de rayon R et de hauteur h en le considérant comme un solide de révolution obtenu par rotation de la droite $x = R$ autour de l'axe y .
- Utilisez la méthode des disques (*disk method*). Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - Utilisez la méthode des tubes (*shell method*). Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - * Utilisez `wxdraw3d` pour produire une représentation graphique du cylindre avec un rayon de 1.
13. Calculez le volume d'une couronne cylindrique de rayon intérieur a , de rayon extérieur b et de hauteur h .
- Utilisez la méthode des rondelles (*washer method*). Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - Utilisez la méthode des tubes (*shell method*). Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - * Utilisez `wxdraw3d` pour produire une représentation graphique de la couronne cylindrique avec un rayon intérieur de 1 et un rayon extérieur de 2. Vous devrez créer trois surfaces paramétriques : le cylindre intérieur, le cylindre extérieur et la « rondelle » qui coiffe l'extrémité.
14. Calculez le volume d'une sphère de rayon R en considérant un hémisphère comme une surface de révolution d'une courbe $f(x)$ définie sur $[0, R]$.
- Utilisez la méthode des disques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - Utilisez la méthode des coquilles cylindriques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - * Utilisez `makelist` et `wxdraw3d` pour générer 20 disques empilés dans un graphique et 20 coquilles cylindriques imbriquées dans un autre graphique, illustrant chaque méthode d'intégration pour une sphère de rayon 1.

15. Calculez le volume d'un cône de hauteur h et de base de rayon R en considérant le cône comme un solide de révolution d'un segment de droite défini sur $[0, R]$.
- (a) Utilisez la méthode des disques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - (b) Utilisez la méthode des coquilles cylindriques. Assurez-vous d'énoncer explicitement l'élément de volume dV avant de sommer les éléments à l'aide d'une intégrale.
 - (c) * Utilisez `makelist` et `wxdraw3d` pour générer 20 disques empilés dans un graphique et 20 coquilles cylindriques imbriquées dans un autre graphique, illustrant chaque méthode d'intégration pour un cône de hauteur 1 et de rayon 1.
16. Calculez l'aire de la surface latérale du cône de l'exemple précédent.
- (a) En utilisant une intégrale en x . Assurez-vous d'énoncer explicitement l'élément d'aire dA avant de sommer les éléments à l'aide d'une intégrale.
 - (b) En utilisant une intégrale en y . Assurez-vous d'énoncer explicitement l'élément d'aire dA avant de sommer les éléments à l'aide d'une intégrale.
 - (c) * Utilisez `wxdraw3d` pour tracer le cône, en incluant un élément d'aire dA .

Corrigé

1 Exercice1


```
(% i1)  f(x) := -x^2 + 2*x;
(%o1)  f(x) := -x^2 + 2x
```

```
(% i2)  inter : solve(f(x)=0,x);
(inter) [x = 0, x = 2]
```

```
(% i4)  x1 : rhs(inter[1]); x2 : rhs(inter[2]);
(x1)  0
(x2)  2
```

La courbe de f est donc une parabole qui est positive sur $[0,2]$ et négative sur $[2,4]$.

```
(% i5) wxdraw2d(
  title= ,
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-1,5],
  yrange=[-8,2],
  fill_color=red,
  filled_func=true,
  filled_func=f(x),
  explicit(0,x,0,2),
  fill_color=blue,
  filled_func=true,
  filled_func=f(x),
  explicit(0,x,2,4),
  filled_func=false,
  color=black,
  line_width=1,
  key= f(x) ,
  explicit(f(x),x,0,4),
  key= ,
  point_type=7,
  color=green,
  point_size=1,
  points([[x1,f(x1)],[x2,f(x2)]])
)
;
```



(%t5)

(%o5)

Il reste à calculer chaque aire par la commande integrale

```
(% i6) Aire : integrate(f(x),x,0,2)+integrate(f(x),x,2,4);
```

```
(Aire)  -  $\left(\frac{16}{3}\right)$ 
```

ou directement sur l'intervalle $[0,4]$

```
(% i7) integrate(f(x),x,0,4);
```

```
(%o7)  $-\left(\frac{16}{3}\right)$ 
```

```
(% i8) float(Aire);
```

```
(%o8) -5.333333333333333
```

2 Exercice 2

```
(% i9) kill(all)$
```


```
(% i2) f(x):=sqrt(1-x^2);g(x):=cos(%pi/2*x);
```

```
(%o1)  $f(x) := \sqrt{1-x^2}$ 
```

```
(%o2)  $g(x) := \cos\left(\frac{\pi}{2}x\right)$ 
```

Tracé des courbes pour visualiser la situation, sachant que f est définie sur [-1,1]

```
(% i3) wxdraw2d(
  title= ,
  proportional_axes=xy,
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-1,1],
  yrange=[0,1.2],
  filled_func=true,
  fill_color=green,
  filled_func=f(x),
  explicit(g(x),x,-1,1),
  filled_func=false,
  color=black,
  line_width=1,
  key= f(x) ,
  explicit(f(x),x,-1,1),
  color=blue,
  key= g(x) ,
  explicit(g(x),x,-1,1)
)
;
```



(%t3)

(%o3)

On montre que les deux fonctions sont paires :

(% i5) $f(-x);g(-x);$

(%o4) $\sqrt{1-x^2}$

(%o5) $\cos\left(\frac{\pi x}{2}\right)$

donc l'aire comprise entre les 2 courbes est égale à 2 fois celle comprise entre 0 et 1. Comme f est toujours au dessus de g, on a donc

(% i6) $\text{Aire}:2*\text{integrate}(f(x)-g(x),x,0,1);$

(Aire) $\frac{\pi^2 - 8}{2\pi}$

(% i7) $\text{float}(\text{Aire});$

(%o7) 0.29755678205973385

3 Exercice 3


(% i8) $\text{kill}(\text{all})\$$

(% i2) $f(x) := \text{product}(x - k, k, -3, 3); g(x) := \exp(x);$

(%o1) $f(x) := \prod_{k=-3}^3 x - k$

(%o2) $g(x) := \exp(x)$

```
(% i3) wxdraw2d(
  title= ,
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[-5,7],
  yrange=[-100,100],
  /* filled_func=true,
  fill_color=green,
  filled_func=f(x),
  explicit(g(x),x,-1,1),
  filled_func=false,*/
  color=black,
  line_width=1,
  key= f(x) ,
  explicit(f(x),x,-10,10),
  color=blue,
  key= g(x) ,
  explicit(g(x),x,-10,10)
)
;
```



```
(%t3)
```

```
(%o3)
```

solve ne fonctionnant pas, on cherche une valeur approchée de chacune des racines sur chaque intervalle

```
(% i4) x1 :find_root(f(x)=g(x),x,-4,-2.5);
(x1) - 2.999930834791456
```

```
(% i5) x2 :find_root(f(x)=g(x),x,-2.5,-2);
(x2) - 2.0011249027269153
```

```
(% i6) x3 :find_root(f(x)=g(x),x,-2,-0.5);
(x3) - 0.992240478447905
```

```
(% i7) x4 :find_root(f(x)=g(x),x,-0.5,1);  
(x4) - 0.027063080088140838
```


```
(% i8) x5 :find_root(f(x)=g(x),x,0.02,1.5);  
(x5) 1.058287348431664
```

```
(% i9) x6 :find_root(f(x)=g(x),x,1.5,3);  
(x6) 1.9370070271566375
```

```
(% i10) x7 :find_root(f(x)=g(x),x,3,4);  
(x7) 3.0268457522423007
```

On peut représenter ces points et l'aire entre les 2 courbes :

```
(% i11) wxdraw2d(  
  title= ,  
  grid=true,  
  xaxis=true,  
  yaxis=true,  
  xrange=[-5,7],  
  yrange=[-100,100],  
  filled_func=true,  
  fill_color=green,  
  filled_func=f(x),  
  explicit(g(x),x,x1,x7),  
  filled_func=false,  
  color=black,  
  line_width=1,  
  key= f(x) ,  
  explicit(f(x),x,-10,10),  
  color=blue,  
  key= g(x) ,  
  explicit(g(x),x,-10,10),  
  color=magenta,  
  point_type=7,  
  point_size=1,  
  key= ,  
  points([[x1,f(x1)],[x2,f(x2)],[x3,f(x3)],[x4,f(x4)],[x5,f(x5)],[x6,f(x6)],[x7,f(x7)]])  
)$
```



(%t11)

(% i12) ratprint :false\$

(% i13) A1 :float(integrate(f(x)-g(x),x,x1,x2));
(A1) 57.206196329481514

(% i14) A2 :float(integrate(f(x)-g(x),x,x2,x3));
(A2) -14.85904535306223

(% i15) A3 :float(integrate(f(x)-g(x),x,x3,x4));
(A3) 7.341161154688287

(% i16) A4 :float(integrate(f(x)-g(x),x,x4,x5));
(A4) -9.770068596698279

(% i17) A5 :float(integrate(f(x)-g(x),x,x5,x6));
(A5) 10.260167577267717

(% i18) A6 :float(integrate(f(x)-g(x),x,x6,x7));
(A6) -70.48963425935528

(% i19) Aire :A1-A2+A3-A4+A5-A6;
(Aire) 169.92627327055334

4 Exercice 4


(% i20) kill(all)\$

```
(% i1) f(x) :=cos(x)$
```

4.1 Rotation de $f(x)$ autour de l'axe des y

```
(% i2) wxdraw2d(  
  xrange=[-1,%pi],  
  yrange=[-1.2,1.2],  
  xaxis=true,  
  yaxis=true,  
  xtics=false,  
  ytics=false,  
  line_width=2,  
  title= f(x)=cos(x) sur  $[0,\pi]$  ,  
  color=black,  
  explicit(f(x),x,0,%pi)  
)$
```

$f(x)=\cos(x)$ sur $[0,\pi]$




```
(%t2)
```

```
(% i3) wxdraw3d(
axis_3d = false,
view = [60, 130],
xrange = [-4, 4],
yrange = [-4, 4],
zrange = [-1.5, 1.5],
xtics = false,
ytics = false,
ztics = false,

/* Axes */
color = black,
line_width = 2,
parametric(t, 0, 0, t, -3.5, 3.5), /* axe x */
parametric(0, t, 0, t, -3.5, 3.5), /* axe y */
parametric(0, 0, t, t, -1.5, 1.5), /* axe z */

/* Solide de révolution : f(x)=cos(x), x in [0,pi], autour de l'axe y */
/* Paramétrisation : X = x*cos(t), Y = cos(x), Z = x*sin(t) */
color = dark_grey,
opacity = 0.7,
parametric_surface( x * cos(t), cos(x), x * sin(t), x, 0, %pi, t, 0, 2 * %pi),
color=black,
parametric_surface( x * cos(t), cos(x), x * sin(t), x, 1.2,1.3, t, 0, 2 * %pi),
color=dark_red,
parametric_surface( 1.3, cos(x), x * sin(t), x, 0,1.3, t, 0, 2 * %pi)
) $
```



(%t3)

```
(% i4) solve(y=cos(x),x);
solve : using arc - trig function to get a solution. Some solutions will be lost.
```

```
(%o4) [x = acos(y)]
```

```
(% i5) dV : %pi * acos(y)^2;
```

$$(dV) = \pi \cos(y)^2$$

Lorsque x varie de 0 à π , $y=\cos(x)$ varie de 1 à -1, donc

```
(% i6) integrate(dV,y,-1,1);
```

```
(%o6)  $\pi (\pi^2 - 4)$ 
```

```
(% i7) float(%);
```

```
(%o7) 18.439906065940644
```

4.2 Méthode des coques cylindriques

```
(% i8) V_coque :integrate(2*%pi*x*abs(f(x)),x,0,%pi);
```

```
(V_coque)  $2\pi \int_0^{\pi} x |\cos(x)| dx$ 
```

```
(% i9) V1 :integrate(2*%pi*x*f(x),x,0,%pi/2);
```

```
(V1)  $2 \left( \frac{\pi}{2} - 1 \right) \pi$ 
```

```
(% i10) V2 :integrate(-2*%pi*x*f(x),x,%pi/2,%pi);
```

```
(V2)  $- \left( 2 \left( - \left( \frac{\pi}{2} \right) - 1 \right) \pi \right)$ 
```

```
(% i11) expand(V1+V2);
```

```
(%o11)  $2\pi^2$ 
```

```
(% i12) float(%);
```


```
(%o12) 19.739208802178716
```

Comme la fonction change de signe sur l'intervalle, la méthode 1 calcule le volume du solide plein, tandis que la méthode 2 considère que le solide est troué par la partie négative de $f(x)$.

```
(% i13) wxdraw3d(
axis_3d = false,
view = [60, 130],
xrange = [-4, 4],
yrange = [-4, 4],
zrange = [-1.5, 1.5],
xtics = false,
ytics = false,
ztics = false,

/* Axes */
color = black,
line_width = 2,
parametric(t, 0, 0, t, -3.5, 3.5), /* axe x */
parametric(0, t, 0, t, -3.5, 3.5), /* axe y */
parametric(0, 0, t, t, -1.5, 1.5), /* axe z */

/* Solide de révolution : f(x)=cos(x), x in [0,pi], autour de l'axe y */
/* Paramétrisation : X = x*cos(t), Y = cos(x), Z = x*sin(t) */
color = dark_grey,
opacity = 0.7,
parametric_surface( x * cos(t), cos(x), x * sin(t), x, 0, %pi, t, 0, 2 * %pi),
color=black,
parametric_surface( x * cos(t), cos(x), x * sin(t), x, 1.2,1.3, t, 0, 2 * %pi),
color=dark_red,
parametric_surface( x * cos(t), cos(x), 1.3 * sin(t), x, 1.2,1.3, t, 0, 2 * %pi)
) $
```



(%t13)


```
(% i14) wxdraw3d(
axis_3d = false,
view = [70, 45], /* Angle de vue surélevé pour bien voir les disques */
xrange = [-4, 4],
yrange = [-4, 4],
zrange = [-1.5, 1.5],
xtics = false,
ytics = false,
ztics = false,

/* Axes (Z est l'axe vertical de révolution) */
color = gray, line_width = 1,
parametric(t, 0, 0, t, -3.5, 3.5), /* axe x */
parametric(0, t, 0, t, -3.5, 3.5), /* axe y */
color = black, line_width = 2,
parametric(0, 0, t, t, -1.5, 1.5), /* axe z (Axe de révolution) */

/* Solide de révolution principal : f(x)=cos(x), x in [0,pi] */
color = light-blue ,
opacity = 0.5,
parametric_surface( x * cos(t), x * sin(t), cos(x), x, 0, %pi, t, 0, 2 * %pi),

/* Illustration de la MÉTHODE DES DISQUES */
/* On dessine un disque horizontal à la hauteur z = 0.5 */
/* Le rayon maximum de ce disque est égal à r_max = acos(0.5) = %pi/3 */
color = dark_red,
opacity = 1,
parametric_surface( r * cos(t), r * sin(t), 0.5, r, 0, acos(0.5), t, 0, 2 * %pi),

/* Une ligne noire pour bien marquer le contour extérieur du disque */
color = black, line_width = 2,
parametric(acos(0.5) * cos(t), acos(0.5) * sin(t), 0.5, t, 0, 2 * %pi)
)$
```



(%t14)


```
(% i15) wxdraw3d(
axis_3d = false,
view = [70, 45],    /* Angle de vue surélevé pour bien voir l'intérieur */
xrange = [-4, 4],
yrange = [-4, 4],
zrange = [-1.5, 1.5],
xtics = false,
ytics = false,
ztics = false,

/* Axes (Z est maintenant l'axe vertical de révolution) */
color = gray, line_width = 1,
parametric(t, 0, 0, t, -3.5, 3.5), /* axe x */
parametric(0, t, 0, t, -3.5, 3.5), /* axe y */
color = black, line_width = 2,
parametric(0, 0, t, t, -1.5, 1.5), /* axe z (Axe de révolution) */

/* Solide de révolution principal : f(x)=cos(x), x in [0,pi] */
/* Le rayon horizontal est x, la hauteur verticale est cos(x) */
color = light-blue ,
opacity = 0.6,
parametric_surface( x * cos(t), x * sin(t), cos(x),  x, 0, %pi,  t, 0, 2 * %pi),

/* Mise en valeur d'une coque cylindrique à x = 1.3 */
/* Cela illustre parfaitement la méthode des coques de votre exercice ! */
color = dark_red,
opacity = 1,
parametric_surface( 1.3 * cos(t), 1.3 * sin(t), z,  t, 0, 2 * %pi,  z, cos(1.3), 1),

/* Ligne noire pour souligner le contour de cette coque */
color = black, line_width = 2,
parametric(1.3 * cos(t), 1.3 * sin(t), cos(1.3),  t, 0, 2 * %pi)
)$
```



(%t15)

5 Exercice 5

5.1 Méthode des disques

(% i16) $dV : \pi f(x)^2;$
(dV) $\pi \cos(x)^2$

(% i17) $V : \text{integrate}(dV, x, 0, \pi);$
(V) $\frac{\pi^2}{2}$

5.2 Méthode des coques cylindriques

$y = \cos(x)$ varie de -1 à 1 quand x varie de 0 à π ; par symétrie du solide, on calcule 2 fois le volume du solide obtenu par rotation de la partie positive, soit y variant de 0 à 1. La hauteur du cylindre vaut $(\pi - \arccos(y)) - \arccos(y) = \pi - 2\arccos(y)$. D'où l'intégrale de volume :

(% i18) $V_coque : \text{integrate}(2 * \pi * y * (\pi - 2 * \arccos(y)), y, 0, 1);$
(V_coque) $\frac{\pi^2}{2}$


```
(% i19) wxdraw3d(
axis_3d = false,
view = [60, 130],
xrange = [-1, 4], /* Ajusté car x va de 0 à pi */
yrange = [-1.5, 1.5], /* Ajusté car le rayon max (cos) est de 1 */
zrange = [-1.5, 1.5], /* Ajusté car le rayon max (cos) est de 1 */
xtics = false,
ytics = false,
ztics = false,

/* Axes */
color = black,
line_width = 2,
parametric(t, 0, 0, t, -1, 3.5), /* axe x */
parametric(0, t, 0, t, -1.5, 1.5), /* axe y */
parametric(0, 0, t, t, -1.5, 1.5), /* axe z */

/* Solide de révolution : f(x)=cos(x), x in [0,pi], autour de l'axe x */
/* Paramétrisation : X = x, Y = cos(x)*cos(t), Z = cos(x)*sin(t) */
color = dark_grey,
opacity = 0.7,
parametric_surface( x, cos(x) * cos(t), cos(x) * sin(t), x, 0, %pi, t, 0, 2 * %pi),

/* Portion mise en valeur (exemple d'une tranche ou d'un disque) */
color = black,
parametric_surface( x, cos(x) * cos(t), cos(x) * sin(t), x, 1.2, 1.3, t, 0, 2 * %pi),

/* Disque de section à une coordonnée x fixe */
color = dark_red,
parametric_surface( 1.3, cos(1.3) * r * cos(t), cos(1.3) * r * sin(t), r, 0, 1, t, 0, 2 * %pi)
)$
```




(%t19)

6 Exercice 6

(% i20) $f(x) := 1 + 0.5 * x$

(% i21) $g(x) := x$

```
(% i22) wxdraw2d(
  title= Tracé de la région ,
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[0,2],
  yrange=[-1,3],
  color=blue,
  line_width=2,
  key= f(x) ,
  explicit(f(x),x,0,2),
  color=red,
  key= g(x) ,
  explicit(g(x),x,0,2),
  key= ,
  filled_func=true,
  fill_color=green,
  filled_func=f(x),
  explicit(g(x),x,0,2)
)
;
```



(%t22)

(%o22)


(% i23) solve(f(x)=g(x),x);

(%o23) $[x = 2]$

Les courbes se croisent en $x=2$, et on a $f(x)$ toujours au dessus de $g(x)$ sur l'intervalle $[0,2]$.

6.1 Rotation autour de l'axe des y

```
(% i24) wxdraw3d(  
axis_3d = false, view = [70, 50],  
xrange = [-2.5, 2.5], yrange = [-2.5, 2.5], zrange = [-0.5, 2.5],  
xtics = false, ytics = false, ztics = false,  
  
/* Axes : Z est l'axe vertical de révolution (Axe Y du problème) */  
color = gray, line_width = 1,  
parametric(t, 0, 0, t, -2, 2), /* axe X */  
parametric(0, t, 0, t, -2, 2), /* axe Z (Axe X du problème) */  
color = black, line_width = 2,  
parametric(0, 0, t, t, -0.5, 2.2), /* axe de révolution */  
  
/* Surface engendrée par  $f(x) = 1 + 0.5x$  */  
color = light-blue, opacity = 0.5,  
parametric_surface(x*cos(t), x*sin(t), 1 + 0.5*x, x, 0, 2, t, 0, 2*%pi),  
  
/* Surface engendrée par  $g(x) = x$  */  
color = light-gray, opacity = 0.6,  
parametric_surface(x*cos(t), x*sin(t), x, x, 0, 2, t, 0, 2*%pi),  
  
/* Base plate supérieure pour fermer le solide */  
color = light-blue, opacity = 0.5,  
parametric_surface(r*cos(t), r*sin(t), 2, r, 0, 2, t, 0, 2*%pi),  
  
/* ÉLÉMENT DE VOLUME : Rondelle à la hauteur  $z = 1.4$  */  
/* Rayon intérieur :  $2*z - 2$  | Rayon extérieur :  $z$  */  
color = red, opacity = 1,  
parametric_surface(r*cos(t), r*sin(t), 1.4, r, 2*1.4 - 2, 1.4, t, 0, 2*%pi),  
color = black, line_width = 1.5,  
parametric((2*1.4-2)*cos(t), (2*1.4-2)*sin(t), 1.4, t, 0, 2*%pi),  
parametric(1.4*cos(t), 1.4*sin(t), 1.4, t, 0, 2*%pi)  
)$
```



(%t24)

(% i25) solve(f(x)=y,x);

(%o25) $[x = 2y - 2]$

La solution est valable uniquement sur $[1,2]$. En effet De $y = 0$ à $y = 1$, le rayon intérieur est 0 (l'axe y) et le rayon extérieur est $x = y$. Le solide est un cône plein, l'élément de volume est $\pi \cdot y^2$. De 1 à 2 le solide a un trou au milieu, avec un rayon intérieur de $2y-2$ et un rayon extérieur de y . L'élément de volume est donc dans ce cas $dV = \pi \cdot (y^2 - (2y-2)^2)$. Donc

(% i26) V1 :integrate(%pi*y^ 2,y,0,1);

(V1) $\frac{\pi}{3}$

(% i27) V2 :integrate(%pi*(y^ 2-(2*y-2)^ 2),y,1,2);

(V2) π

(% i28) V :V1+V2;

(V) $\frac{4\pi}{3}$

6.2 Rotation autour de l'axe des x

```
(% i29) wxdraw3d(
axis_3d = false, view = [75, 40],
xrange = [-2.5, 2.5], yrange = [-2.5, 2.5], zrange = [-0.5, 2.5],
xtics = false, ytics = false, ztics = false,


/* Axes : Z est l'axe vertical de révolution (Axe X du problème) */
color = gray, line_width = 1,
parametric(t, 0, 0, t, -2, 2), /* axe Y */
parametric(0, t, 0, t, -2, 2), /* axe Z (Axe Y du problème) */
color = black, line_width = 2,
parametric(0, 0, t, t, -0.5, 2.2), /* axe de révolution */

/* PAROI EXTÉRIEURE : engendrée par  $f(z) = 1 + 0.5*z$  */
color = light-blue, opacity = 0.5,
parametric_surface((1 + 0.5*z)*cos(t), (1 + 0.5*z)*sin(t), z, z, 0, 2, t, 0, 2*%pi),

/* PAROI INTÉRIEURE : engendrée par  $g(z) = z$  */
color = light-gray, opacity = 0.7,
parametric_surface(z*cos(t), z*sin(t), z, z, 0, 2, t, 0, 2*%pi),

/* Bordure supérieure nette (Le cercle de rencontre en  $z = 2$ ) */
color = dark_blue, line_width = 2,
parametric(2*cos(t), 2*sin(t), 2, t, 0, 2*%pi),

/* ÉLÉMENT DE VOLUME : Rondelle à la hauteur  $z = 1.2$  (c'est-à-dire à  $x = 1.2$ ) */
/* Rayon intérieur : 1.2 | Rayon extérieur :  $1 + 0.5*1.2 = 1.6$  */
color = red, opacity = 1,
parametric_surface(r*cos(t), r*sin(t), 1.2, r, 1.2, 1.6, t, 0, 2*%pi),
color = black, line_width = 1.5,
parametric(1.2*cos(t), 1.2*sin(t), 1.2, t, 0, 2*%pi),
parametric(1.6*cos(t), 1.6*sin(t), 1.2, t, 0, 2*%pi)
)$
```



(%t29)

Pour x variant de 0 à 2, on a rayon intérieur de $g(x)$ et le rayon extérieur égal à $f(x)$

```
(% i30) V3 :integrate(%pi*(f(x)^ 2-g(x)^ 2),x,0,2);
(V3) 2π
```

7 Exercice 7

```
(% i31) kill(all)$
```

```
(% i1) f(x) :=sin(x)$
```

```
(% i2) g(x) :=cos(x)$
```

```
(% i3) x1 :find_root(f(x)=g(x),x,0,2);
(x1) 0.7853981633974483
```

```
(% i4) x2 :find_root(f(x)=g(x),x,2,4);
(x2) 3.9269908169872414
```


Valeurs de y correspondantes à x1 et x2 (on doit intégrer sur y pour les rondelles)

```
(% i6) y1 :f(x1);y2 :f(x2);
(y1) 0.7071067811865475
```

```
(y2) -0.7071067811865475
```

```
(% i7) wxdraw2d(
title= Tracé de la région ,
grid=true,
xaxis=true,
yaxis=true,
xrange=[0,3*%pi/2],
yrange=[-1.5,2],
color=blue,
line_width=2,
key= f(x) ,
explicit(f(x),x,0,3*%pi/2),
color=red,
key= g(x) ,
explicit(g(x),x,0,3*%pi/2),
key= ,
filled_func=true,
fill_color=green,
filled_func=f(x),
explicit(g(x),x,x1,x2)
)$
```

Tracé de la région



(%t7)

7.1 Méthode des rondelles

$V = \pi \int (x_{\text{exterieur}}^2 - x_{\text{interieur}}^2) dy$ Le calcul des intervalles sur lesquels on peut trouver x en fonction de y est trop complexe dans ce cas (pas de bijectivité de la fonction sinus). La méthode des rondelles n'est pas adaptée au problème posé.


7.2 Méthode des coques cylindriques

```
(% i12) /* Méthode des coques cylindriques */
         integrand_coques : x * (sin(x) - cos(x));
         V_coques : 2 * %pi * integrate(integrand_coques, x, x1,x2);
         V_coques_simp : radcan(V_coques);
         print( Volume (coques) = , V_coques_simp)$
         print( Volume (coques) = , float(V_coques_simp))$
(integrand_coques)  x (sin (x) - cos (x))
(V_coques)          2572524314397542π
                   193007134496915
(V_coques_simp)     2572524314397542π
                   193007134496915
Volume (coques) = 2572524314397542π
                  193007134496915
Volume (coques) = 41.87318519783327
```

7.3 Représentation graphique

On génère les solides obtenus par la révolution de $\sin(x)$ et $\cos(x)$ autour de l'axe des y sur l'intervalle.

```
(% i13) wxdraw3d(
/* Surface engendrée par sin(x) */
color = red,
parametric_surface(
x * cos(t), sin(x), x * sin(t),
x, %pi/4, 5*%pi/4,
t, 0, 2*%pi
),
/* Surface engendrée par cos(x) */
color = blue,
parametric_surface(
x * cos(t), cos(x), x * sin(t),
x, %pi/4, 5*%pi/4,
t, 0, 2*%pi
),
title = Solide de revolution par rotation de la region autour de y ,
xlabel = X , ylabel = Y , zlabel = Z ,
surface_hide = true
)$
```




(%t13)

8 Exercice 8

```
(% i21) f(x) :=sin(x)$
SLOPE(a,b,c,d) :=(d-b)/(c-a)$
LINE(a,b,c,d) :=SLOPE(a,b,c,d)*(x-a)+b$
LENGTH(a,b,c,d) :=sqrt((c-a)^ 2+(d-b)^ 2)$
delx(n) :=(2*%pi-0)/n$
x(n,i) :=0+(i)*delx(n)$
SEGMENTS(n) :=makelist(explicit(LINE(x(n,i),f(x(n,i))),x(n,i+1),f(x(n,i+1))),x,x(n,i),x(n,i+1)),i,0,n-1)$
APPROX(n) :=float(sum(LENGTH(x(n,i),f(x(n,i))),x(n,i+1),f(x(n,i+1))),i,0,n-1))$
```

```
(% i22) represente(k) := wxdraw2d(
  grid=true,
  xaxis=true,
  yaxis=true,
  xrange=[0,2*%pi],
  yrange=[-2,2],
  title=concat( Pour n= ,k, longueur arc = ,APPROX(k)),
  color=black,
  explicit(f(x),x,0,2*%pi),
  line_type=dots,
  line_width=3,
  color=red,
  SEGMENTS(k)
)$
```


```
(% i23) represente(4)$
```



```
(%t23)
```

```
(% i24) represente(8)$
```


Pour $n=8$ longueur arc = 7.580182617061441



(%t24)

(% i25) `represente(16)`


Pour $n=16$ longueur arc = 7.625057365834933



(%t25)

(% i26) `represente(32)`


Pour n=32 longueur arc =7.636549929994366



(%t26)

(% i27) represente(64)\$

Pour n=64 longueur arc =7.639433473994346



(%t27)

(% i28) 'integrate(sqrt(1+diff(f(x),x)^ 2),x,0,2*%pi)=quad_qags(sqrt(1+diff(f(x),x)^ 2),x,0,2*%pi)[1];

(%o28) $\int_0^{2\pi} \sqrt{\cos(x)^2 + 1} dx = 7.640395578055423$

Comparaison des résultats et calcul du pourcentage

(% i29) 100*((quad_qags(sqrt(1+diff(f(x),x)^ 2),x,0,2*%pi)[1]-APPROX(64))/(quad_qags(sqrt(1+diff(f(x),x)^ 2),x,0,2*%pi)[1]));


(%o29) 0.012592333096468018

9 Exercice 9

```
(% i30) kill(all)$
```


```
(% i1) f(x) := exp(-0.2*x)*cos(3*x);  
(%o1)  $f(x) := \exp(-0.2x) \cos(3x)$ 
```

```
(% i2) wxdraw2d(  
  xrange=[-1,11],  
  yrange=[-2,2],  
  xaxis=true,  
  yaxis=true,  
  line_width=2,  
  title= f(x) sur [-0,10] ,  
  color=black,  
  explicit(f(x),x,0,10)  
)$
```



```
(%t2)
```

```
(% i3) /* Surface paramétrée : rotation de f(x) autour de l'axe x */  
/* x = x, y = f(x)*cos(t), z = f(x)*sin(t) */  
wxdraw3d(  
  title = Surface de révolution de f(x) autour de Ox ,  
  xlabel = x , ylabel = y , zlabel = z ,  
  surface_hide = true,  
  color = blue,  
  parametric_surface(  
    x,  
    exp(-0.2*x) * cos(3*x) * cos(t),  
    exp(-0.2*x) * cos(3*x) * sin(t),  
    x, 0, 10,  
    t, 0, 2*%pi  
  ),  
  grid = [50, 50]  
)$
```



(%t3)

(% i4) `S : 2*%pi*integrate(abs(f(x))*sqrt(1+diff(f(x),x)^ 2),x,0,10 );`

$$(S) \quad 2\pi \int_0^{10} e^{-(0.2x)} \sqrt{\left(-\left(3e^{-(0.2x)} \sin(3x) \right) - 0.2e^{-(0.2x)} \cos(3x) \right)^2 + 1} |\cos(3x)| \, dx$$

On utilise le calcul numérique d'une intégrale

(% i5) `Sn : 2*%pi*quad_qags(abs(f(x))*sqrt(1+diff(f(x),x)^ 2),x,0,10 );`


(Sn) `[7.740897980831299π, 4.812565368439436510-8π, 10710π, 0]`

(% i6) `float(Sn[1]);`

(%o6) `24.318748228767674`

10 Exercice 10

```
(% i7) wxdraw3d(
title      = Surface de révolution de f(x) autour de Oy ,
xlabel     = x , ylabel = y , zlabel = z ,
surface_hide = true,
color      = blue,
parametric_surface(
x * cos(t), /* coordonnée X */
exp(-0.2*x)*cos(3*x), /* hauteur Y = f(x) */
x * sin(t), /* coordonnée Z */
x, 0, 10,
t, 0, 2*%pi
),
grid = [60, 40]
)$
```



(%t7)

```
(% i8) S2 :2*%pi*integrate(x*sqrt( 1+ diff(f(x),x)^ 2 ),x,0,10 );
```

$$(S2) \quad 2\pi \int_0^{10} x \sqrt{\left(-\left(3e^{-(0.2x)} \sin(3x) \right) - 0.2e^{-(0.2x)} \cos(3x) \right)^2 + 1} dx$$

```
(% i9) S2n :S2 :2*%pi*quad_qags(x*sqrt( 1+ diff(f(x),x)^ 2 ),x,0,10 );
```

$$(S2n) \quad [120.1936682311244\pi, 6.21332708717484310^{-7}\pi, 1302\pi, 0]$$

```
(% i10) float(S2n[1]);
```

```
(%o10) 377.5995451229093
```

11 Exercice 11

```
(% i11) kill(all)$
```

On considère un disque de centre O et de rayon R. Equation du cercle frontière :

```
(% i1)  assume(R>0);  
(%o1)  [R > 0]
```

```
(% i2)  eqc :x^2+y^2=R^2;  
(eqc)   $y^2 + x^2 = R^2$ 
```

```
(% i3)  solve(eqc,y);  
(%o3)   $\left[ y = -\sqrt{R^2 - x^2}, y = \sqrt{R^2 - x^2} \right]$ 
```

On définit le demi cercle supérieur

```
(% i4)  eqcs :solve(eqc,y)[2];  
(eqcs)   $y = \sqrt{R^2 - x^2}$ 
```

```
(% i5)  rhs(eqcs);  
(%o5)   $\sqrt{R^2 - x^2}$ 
```

```
(% i6)  aire :2*integrate(rhs(eqcs),x,-R,R);  
(aire)   $\pi R^2$ 
```

Longueur d'un anneau $2*\pi*r$, qu'on multiplie par dr avant de sommer

```
(% i7)  aire2 :integrate(2*%pi*r,r,0,R);  
(aire2)   $\pi R^2$ 
```

12 Exercice 12

```
(% i8)  kill(all)$
```

12.1 Méthode des disques


```
(% i1)  V_disque : integrate(%pi * R^2, y, 0, h);  
(V_disque)   $\pi R^2 h$ 
```

12.2 Méthode des tubes

```
(% i2)  V_tube : integrate(2 * %pi * x * h, x, 0, R);  
(V_tube)   $\pi R^2 h$ 
```

12.3 Représentation graphique

```
(% i3) wxdraw3d(  
/* Cylindre lateral */  
color = blue,  
parametric_surface(  
cos(t), sin(t), z,  
t, 0, 2*%pi,  
z, 0, 1  
),  
/* Disque superieur */  
color = red,  
parametric_surface(  
r * cos(t), r * sin(t), 1,  
t, 0, 2*%pi,  
r, 0, 1  
),  
/* Disque inferieur */  
color = red,  
parametric_surface(  
r * cos(t), r * sin(t), 0,  
t, 0, 2*%pi,  
r, 0, 1  
)  
)$
```



(%t3)

13 Exercice 13

```
(% i4) kill(all)$
```

13.1 Méthode des rondelles

```
(% i1) V_rondelle : integrate(%pi * (b^2 - a^2), y, 0, h);  
(V_rondelle)  $\pi (b^2 - a^2) h$ 
```

13.2 Méthode des tubes

```
(% i2) V_tube : integrate(2 * %pi * x * h, x, a, b);  
(V_tube)  $2\pi \left( \frac{b^2}{2} - \frac{a^2}{2} \right) h$ 
```


```
(% i3) expand(V_tube);  
(%o3)  $\pi b^2 h - \pi a^2 h$ 
```

```
(% i4) expand(V_rondelle);  
(%o4)  $\pi b^2 h - \pi a^2 h$ 
```

On obtient bien le même résultat

13.3 Représentation graphique

```
(% i5) wxdraw3d(  
/* Cylindre interieur (rayon 1) */  
color = blue,  
parametric_surface(  
cos(t), sin(t), z,  
t, 0, 2*%pi,  
z, 0, 1  
),  
  
/* Cylindre exterieur (rayon 2) */  
color = green,  
parametric_surface(  
2*cos(t), 2*sin(t), z,  
t, 0, 2*%pi,  
z, 0, 1  
),  
  
/* Rondelle superieure (z = 1) */  
color = red,  
parametric_surface(  
r*cos(t), r*sin(t), 1,  
t, 0, 2*%pi,  
r, 1, 2  
),  
  
/* Rondelle inferieure (z = 0) */  
color = red,  
parametric_surface(  
r*cos(t), r*sin(t), 0,  
t, 0, 2*%pi,  
r, 1, 2  
)  
)$
```



(%t5)

14 Exercice 14

```
(% i6) kill(all)$
```

La courbe qui génère l'hémisphère par rotation est définie par $f(x)$ avec x variant de 0 à R .

```
(% i1) f(x) := sqrt(R^2 - x^2);
```

```
(%o1)  $f(x) := \sqrt{R^2 - x^2}$ 
```

14.1 Méthode des disques

Chaque disque a pour rayon $f(x)$ et pour aire :

```
(% i2) dV_disque : %pi * f(x)^2;
```

```
(dV_disque)  $\pi (R^2 - x^2)$ 
```

On précise que le rayon R est positif pour que l'intégrale se calcule :

```
(% i3) assume(R>0)$
```

```
(% i4) V_hemisphere : integrate(dV_disque, x, 0, R);
```

```
(V_hemisphere)  $\frac{2\pi R^3}{3}$ 
```

```
(% i5) V_sphere : 2 * V_hemisphere;
```

```
(V_sphere)  $\frac{4\pi R^3}{3}$ 
```


```
(% i8) R : 1$
```

```
n : 20$
```

```
dx : R / n$
```

```
(% i9) disques : makelist(  
  parametric_surface(  
    sqrt(R^2 - (k * dx)^2) * cos(u), /* x */  
    sqrt(R^2 - (k * dx)^2) * sin(u), /* y */  
    k * dx + v * dx, /* z varie entre k*dx et (k+1)*dx */  
    u, 0, 2 * %pi,  
    v, 0, 1  
  ),  
  k, 0, n - 1  
)$
```

```
(% i10) apply(wxdraw3d,  
  append(  
    [color = blue, nticks = 30,  
     xlabel = x, ylabel = y, zlabel = z ,  
     title = Methode des disques ],  
    disques  
  )  
)$
```



(%t10)

14.2 Méthode des coques cylindriques

(% i2) `kill(all)$f(x) := sqrt(R^2 - x^2)$assume(R>0)$`

(% i3) `dV_coquille : 2 * %pi * x * f(x);`

$$(dV_coquille) \quad 2\pi x \sqrt{R^2 - x^2}$$

(% i4) `V_hemisphère : integrate(dV_coquille, x, 0, R);`

$$(V_hemisphere) \quad \frac{2\pi R^3}{3}$$


(% i5) `V_sphere : 2 * V_hemisphère;`

$$(V_sphere) \quad \frac{4\pi R^3}{3}$$

(% i8) `R : 1$`
`n : 20$`
`dr : R / n$`

(% i9) `coquilles : makelist(`
`parametric_surface(`
`(k*dr) * cos(t),`
`(k*dr) * sin(t),`
`s,`
`t, 0, 2*%pi,`
`s, 0, sqrt(R^2 - (k*dr)^2)`
`),`
`k, 1, n`
`)$`

```
(% i10) apply(wxdraw3d,
append(
[ color = red, nticks = 30,
xlabel = x , ylabel = y , zlabel = z ,
title = Methode des coquilles cylindriques ],
coquilles
)
)$
```



```
(%t10)
```

On remarque qu'avec 20 coquilles ou disques, le graphique ne permet pas de voir le découpage effectué, mais que l'on retrouve bien une demi-sphère par ces méthodes.

15 Exercice 15

```
(% i11) kill(all)$
```

15.1 Méthode des disques

On exprime le rayon en fonction de z , car la rotation se fait autour de l'axe (Oz) :

```
(% i1) f(x) := h * (1 - x/R);
```

```
(%o1)  $f(x) := h \left(1 - \frac{x}{R}\right)$ 
```

```
(% i2) rayon : solve(z=f(x),x);
```

```
(rayon)  $\left[ x = - \left( \frac{Rz - Rh}{h} \right) \right]$ 
```

```
(% i3) rayon2 : expand(rhs(rayon[1]));
```

```
(rayon2)  $R - \frac{Rz}{h}$ 
```

```
(% i7)  r(z) := R * (1 - z/h);
        assume(R > 0, h > 0)$
        dV_disque : %pi * r(z)^2;
        V_cone : integrate(dV_disque, z, 0, h);
(%o4)  r(z) := R (1 - z/h)
(dV_disque)  pi R^2 (1 - z/h)^2
(V_cone)  pi R^2 h / 3
```


15.2 Méthode des coques

```
(% i9)  dV_coquille : 2 * %pi * x * f(x);
        V_cone : integrate(dV_coquille, x, 0, R);
(dV_coquille)  2 pi h x (1 - x/R)
(V_cone)  pi R^2 h / 3
```

```
(% i11)  n : 20$
        dx : 1/n$
```

```
(% i12)  disques : makelist(
        parametric_surface(
        (1 - k*dx) * cos(u), /* x */
        (1 - k*dx) * sin(u), /* y */
        k*dx + v*dx, /* z varie de k*dx à (k+1)*dx */
        u, 0, 2*%pi,
        v, 0, 1
        ),
        k, 0, n-1
        )$
```

```
(% i13)  apply(wxdraw3d,
        append(
        [color = blue,
        xlabel = x , ylabel = y , zlabel = z ,
        title = Methode des disques - cone ],
        disques
        )
        )$
```




(%t13)

(% i15) n : 20\$
dr : 1/n\$

(% i16) coquilles : makelist(
parametric_surface(
(k*dr) * cos(u), /* x */
(k*dr) * sin(u), /* y */
v * (1 - k*dr), /* z varie de 0 à la hauteur 1 - k*dr */
u, 0, 2*%pi,
v, 0, 1
) ,
k, 1, n
)\$

(% i17) apply(wxdraw3d,
append(
[color = red,
xlabel = x , ylabel = y , zlabel = z ,
title = Methode des coquilles cylindriques - cone],
coquilles
)
)\$



16 Exercice 16

(% i18) kill(all)\$

16.1 Calcul avec une intégrale en x

(% i6) assume($R > 0, h > 0$)\$
 $f(x) := h * (1 - x/R);$
 $ds : \text{sqrt}(1 + \text{diff}(f(x), x)^2);$
 $dA_x : 2 * \%pi * x * ds;$
 $A_x : \text{integrate}(dA_x, x, 0, R);$
 $\text{radcan}(A_x);$

(%o2) $f(x) := h \left(1 - \frac{x}{R}\right)$

(ds) $\sqrt{\frac{h^2}{R^2} + 1}$

(dA_x) $2\pi\sqrt{\frac{h^2}{R^2} + 1}x$

(A_x) $\pi R^2\sqrt{\frac{h^2}{R^2} + 1}$

(%o6) $\pi R\sqrt{h^2 + R^2}$

16.2 Calcul avec une intégrale en y

On utilise l'intégrale en z par rotation

```
(% i12) assume(R > 0, h > 0)$
      g(z) := R * (1 - z/h);
      ds : sqrt(1 + diff(g(z), z)^2);
      dA_z : 2 * %pi * g(z) * ds;
      A_z : integrate(dA_z, z, 0, h);
      radcan(A_z);
```

```
(%o8)   $g(z) := R \left(1 - \frac{z}{h}\right)$ 
```

```
(ds)   $\sqrt{\frac{R^2}{h^2} + 1}$ 
```

```
(dA_z)   $2\pi R \sqrt{\frac{R^2}{h^2} + 1} \left(1 - \frac{z}{h}\right)$ 
```

```
(A_z)   $\pi R \sqrt{\frac{R^2}{h^2} + 1} h$ 
```


```
(%o12)   $\pi R \sqrt{h^2 + R^2}$ 
```

16.3 Visualisation

```
(% i13) cone : parametric_surface(
      x * cos(u),
      x * sin(u),
      1 - x,      /* f(x) = 1 - x avec h=R=1 */
      x, 0, 1,
      u, 0, 2 * %pi
    )$
```

```
(% i16) x0 : 0.5$
      dx : 0.08$
      bande : parametric_surface(
      (x0 + v * dx) * cos(u),
      (x0 + v * dx) * sin(u),
      1 - (x0 + v * dx),
      u, 0, 2 * %pi,
      v, 0, 1
    )$
```

```
(% i17) wxdraw3d(
      color = light_blue, transparent = true, opacity = 0.5,
      cone,
      color = red, opacity = 0.9,
      bande,
      xlabel = x , ylabel = y , zlabel = z ,
      title = Cone et element d aire dA
    )$
```



(%t17)